

Inverse kinematics in Gosia

Nigel Warr

December 12, 2008

1 Understanding what the problem really is

It seems that we did not properly understand the problem that Gosia had with inverse kinematics.

In all that follows, I will consider a test case of 193.6 MeV beam of ^{88}Kr on a 2.1 mg/cm² thick ^{12}C target with a compact-disc-shaped particle detector which subtends the angles from 16.4° to 53.0° in the laboratory system.

Performing an energy loss calculation we find that we lose 103.3 MeV in the target, giving an exit energy of 90.3 MeV.

The first excited state in ^{88}Kr is at 0.7753 MeV.

Figure 1 shows how the angle of the scattered ^{88}Kr and the recoiling ^{12}C depend on each other. We can see that the scattered ^{88}Kr nuclei of the beam have a maximum angle of 7.532°. As the particle detector has a hole from 0 to 16.4°, **no** ^{88}Kr nuclei will be detected.

So for Gosia, in OP,GOSI under EXPT, we must specify Z_1 as 36 and A_1 as 88 to indicate that we are investigating ^{88}Kr and Z_n as -6 and A_n as 12 to indicate that the other nucleus is ^{12}C and we are interested in projectile excitations. The value of θ_{LAB} must then be negative to indicate the target nuclei are detected.

In OP,INTG, we need to specify the integration limits over energy as 90.3 to 193.6 MeV and the theta limits as 16.4° to 53.0°, corresponding to the limits of the particle detector.

So what do we give as meshpoints? For energy, it is clear. We select a range a little bit wider than 90.3 to 193.6 MeV, say from 85 to 205 MeV in 15 MeV steps. For theta it is more complicated. Gosia wants the projectile angle in the laboratory frame not the target angle and it has to be negated (actually, I don't think the negation matters, because the code seems to take the sign from θ_{LAB}




Figure 1: Correspondence between the angle of the scattered projectile and that of the recoiling target nucleus in the laboratory frame.

given in EXPT). This is where we come to the problem...

The angle of the scattered projectile in the laboratory frame is NOT well defined when recoiling target nuclei are detected.

For any given target recoil angle in the laboratory frame, the corresponding projectile scattering angle *depends on the beam energy*, but we are integrating over this beam energy, so we have to evaluate it at several different energies!

Figure 2 shows the same as Fig. 1 but for the energies 85 MeV and 205 MeV, corresponding to the extrema of the meshpoints and only within the range of 16.4° to 53.0° corresponding to the angular range of the particle detector.

The fact that the curves shown in Fig. 2 are double-valued is not really a problem as the events for which the projectile scattering angle is less than a degree correspond to very low target recoil energies (only a few hundred keV) so they will not be detected. So for now at least, we will only consider the other solution ranging from a scattered projectile angle of 4.64° at the inner edge of the particle detector to a maximum of 7.53° about $2/3$ of the way across the particle detector and 6.77° at the outer edge at 85 MeV, with slightly higher values at 205 MeV.




Figure 2: Correspondence between the angle of the scattered projectile and that of the recoiling target nuclei in the laboratory frame at two different energies (85 and 205 MeV) showing just the range of target angle, which corresponds to the particle detector.

Note that the angles to the left of the maximum correspond to higher angles of the projectile scattering angle in the centre of mass frame, which can be accessed only with $IKIN = 0$, while angles to the right of the maximum correspond to lower angles of the projectile scattering angle in the centre of mass frame, which can be accessed only with $IKIN = 1$.

So with the current version of Gosia:

It is not possible to integrate over the whole particle detector in a single integration.

Instead, it has to be broken up into two separate pieces. So at 85 MeV, we could integrate from over target recoil angles of 16.4° to about 40.24° with $IKIN = 0$ and from 40.24° to 53° with $IKIN = 1$ and then add up the results to get the whole integral. But at 205 MeV, the maximum is not at 40.24° but at 40.75° , so the angle for which we need to change from $IKIN = 0$ to $IKIN = 1$ is different. This means that it is not possible to integrate between 40.24° and 40.75° at all.

However, Gosia does not want the angles as target recoil angles, but as scattered projectile angles. So at 85 MeV, we have to integrate with $IKIN = 0$ over projectile scattering angles from 4.64° to 7.53° and with $IKIN = 1$ from 6.77°

to 7.53° . However, for 205 MeV, 4.72° to 7.71° and from 7.03° to 7.71° .

Now, Gosia automatically takes care of the calculation of these integration limits, but we still have to set meshpoints. Since Gosia can only *interpolate* and has problems with extrapolation, we need to have meshpoints which go beyond the limits we want, or at very least up to them. Yet, we only have one set of meshpoints, which has to work for all energies. So what happens, if we set meshpoints from 4.64° to 7.71° for $\text{IKIN} = 0$. These points exactly span the region which we need over the full range of energy. For 205 MeV, this is fine, since 4.64° is lower than we need, but it is perfectly valid and 7.71° is the maximum of the curve. However, for lower energies, this is no longer the case. For example, at 85 MeV, there is no solution to the calculation of the centre of mass angle, which gives a projectile scattering angle of 7.71° , since the maximum scattering angle is 7.53° . This means that the highest meshpoint we can set is 7.53° , so we are forced to set our meshpoints from 4.64° to 7.53° . i.e. exactly the range for the lowest energy meshpoint. Similarly for the case of $\text{IKIN} = 1$, we cannot set a meshpoint at 7.71° , but have to use 7.53° instead, so we set meshpoints from 6.77° to 7.53° .




Figure 3: Correspondence between the angle of the scattered projectile and that of the recoiling target nucleus in the laboratory frame at two different energies (85 and 205 MeV) showing just the range of target angle, which corresponds to the particle detector. The gaps in the curves correspond to regions of integration, which are not accessible and cannot be calculated.

The result of these effects is that there are regions, which are inaccessible to the calculation. At 205 MeV, there is a gap corresponding to target recoil

angles of 34.7° to 47.0° . At 85 MeV this effect is less dramatic as we only have the gap from 40.24° to 40.75° due to the change in the place where IKIN flips over. This effect is illustrated in Fig. 3, where the red and green curves correspond to two energies with $IKIN = 0$ and the blue and purple curves to the same energies with $IKIN = 1$. The fact that the curves of a given energy do not meet shows that we have values of the recoiling target nuclei angle, which cannot be accessed in the calculation.

Moreover, since most of the cross section comes from the higher energies, this hole in the 205 MeV range is fatal.

There are crucial regions over which we cannot integrate at all.

The only workaround is to perform the integration over the energy loss in the target manually, by dividing the target into virtual layers and performing the calculation for each layer with a single energy. This is particularly tedious as different limits and different meshpoints need to be set up for each energy. Extreme care has to be taken by the user, because using a limit which is too low results in missing cross-section and one which is too high results in Gosia aborting with an error.

2 The suggested solution and why it won't work

The solution suggested so far is to modify Gosia to do everything in the centre of mass system and to modify Gosia to allow both IKIN values to be used in one calculation. So what would this mean for Gosia?

Changing everything over to the centre of mass system, requires deep changes throughout Gosia. It is highly non-trivial. Every single formula needs to be checked as to whether it changes when passing from the centre of mass system to the laboratory system. A modification of this order is so complicated, that it would be better to completely rewrite Gosia from scratch, than to undertake such a risky change with the existing code. Moreover, it does not solve the problem. The centre of mass angle, whether for the recoiling target or the scattered projectile nuclei (since the two just differ by π), is undefined in this problem, because the transformation from the laboratory frame of the detected particle to the centre of mass angle **depends on the beam energy** which we are integrating over. So the appropriate limits in the centre of mass system change with the energy loss in the target.

The only frame, in which it makes any sense to set limits, is the laboratory frame of the detected particle.

This is because this is the frame of reference of the particle detector itself. If the particle detector detects scattered projectiles, we need to set the limits in the laboratory frame of the scattered particles. In this frame, the angles for which a scattered projectile can hit the particle detector do not depend on the energy of the beam. Similarly if the particle detector detects target recoils, we need to set the limits in the laboratory frame of the recoiling target nuclei. In this frame, the angles for which a target nucleus can recoil into the particle detector do not depend on the energy of the beam.

Gosia requires the angular integration limits to be in the laboratory frame of the detected particle. This is correct. It is the only thing that makes sense.

However, Gosia does not follow this convention for the meshpoints. Instead it requires the meshpoints in the laboratory frame of the scattered projectile, regardless of which particle is detected. For the case where projectiles are detected, this is fine, of course, but when target nuclei are detected, it results in the problem of meshpoints that cannot be set correctly.

The suggestion to have Gosia integrate over both $IKIN = 0$ and $IKIN = 1$ values is feasible, but would entail quite a bit of modification to OP,INTG. The user would have to specify $IKIN = 3$ or 4 and for $IKIN = 4$ the input would be different. Then these values would have to be preserved, while the usual routines are called with $IKIN = 0$ or 1 . This means an extra loop over $IKIN = 0$ and 1 and an extra pair of limits for the case of $IKIN = 4$ and some extra logic. That is not a trivial modification, but it is possible. However, it would only close up the small gap at 85 MeV between 40.24° and 40.75° and do nothing to address the large gap at the higher energies.

3 What should we do?

From the above, we have learnt the following things:

- The only frame of reference we should be using is the laboratory frame of the detected particles.
- Gosia does it right for the integration limits.
- Gosia does it right for the meshpoints, if the scattered projectiles are detected.
- Gosia does the wrong thing, if the recoiling target nuclei are detected. This is the crux of the problem.

From this, the obvious thing to do is to give up the convention that the meshpoints are specified for the scattered projectile and instead use laboratory angles for the detected particle.

Then we must convert these angles into the scattered projectile's laboratory frame for each energy and pass that value to the rest of Gosia, which would be unchanged. This can be implemented as a single routine, which is called inside both the energy and theta meshpoint loops, where the value tta is normally set to $SIGN(YV(ktt),tth)$ where YV are the angles read from the input file, ktt is the index of the current theta value being evaluated, and tth is the value of θ_{LAB} set in EXPT. This means that tta takes the value of $YV(ktt)$ but the sign of θ_{LAB} . If instead, we require YV to be angles in the detected particles laboratory frame, this is the point where we can convert from YV in the detected particle's laboratory frame to tta in the scattered projectile's laboratory frame, if these frames differ.

4 How do we convert between recoiling target and scattered projectile laboratory frames?

The problem corresponds to looking up the target recoil angle on Fig. 1 and finding the corresponding scattered projectile angle. As we can see, there are two solutions in general. However, the lower value will generally correspond to target nuclei with too little energy to be detected.

Let $\theta_{p_{CM}}$ be the angle of the scattered projectile in the centre of mass frame and $\theta_{p_{Lab}}$ be the equivalent in the laboratory frame, while $\theta_{t_{CM}}$ and $\theta_{t_{Lab}}$ are the equivalents for the target recoils.

We can write:

$$\tan \theta_{p_{Lab}} = \frac{\sin \theta_{p_{CM}}}{\tau + \cos \theta_{p_{CM}}} \quad (1)$$

and

$$\tan \theta_{t_{Lab}} = \frac{\sin \theta_{t_{CM}}}{\tau_p + \cos \theta_{t_{CM}}} \quad (2)$$

where

$$\tau = \tau_p \frac{M_p}{M_t} \quad (3)$$

and

$$\tau_p = \sqrt{\frac{E_p}{E_{p_{min}}}} \quad (4)$$

where

$$E_{p_{min}} = E_p - E_x \times a_{red} \quad (5)$$

and the reduced mass a_{red} is given by

$$a_{red} = 1 + \frac{M_p}{M_t} \quad (6)$$

and E_p is the beam energy, E_x is the energy of the excited state indicated by the parameter NCM (by default the first excited state), and M_p and M_t are the projectile and target nuclei masses in AMU, respectively.

Since

$$\theta_{p_{CM}} = \pi - \theta_{t_{CM}} \quad (7)$$

we can substitute in Eq. 2 and get

$$\tan \theta_{t_{Lab}} = \frac{\sin \theta_{p_{CM}}}{\tau_p - \cos \theta_{p_{CM}}} \quad (8)$$

So the problem is one of inverting Eq. 8 to calculate $\theta_{p_{CM}}$ for a given $\theta_{t_{Lab}}$ and then substituting this value of $\theta_{p_{CM}}$ in Eq. 1 to obtain $\theta_{p_{Lab}}$, as required.

Let $x = \cos \theta_{p_{CM}}$ and $y = \tan \theta_{t_{Lab}}$. Then we have:

$$y = \frac{\sqrt{1-x^2}}{\tau_p - x} \quad (9)$$

or

$$y\tau_p - xy = \sqrt{1-x^2} \quad (10)$$

Squaring both sides we get a quadratic equation in x:

$$x^2(1+y^2) - 2\tau_p y^2 x + \tau_p^2 y^2 - 1 = 0 \quad (11)$$

Solving for x we get:

$$x = \frac{\tau_p y^2 \pm \sqrt{\tau_p^2 y^4 - (1+y^2)(\tau_p^2 y^2 - 1)}}{(1+y^2)} \quad (12)$$

Once we have $x = \cos \theta_{p_{CM}}$, we can substitute into Eq. 1.

$$\theta_{p_{Lab}} = \arctan \frac{\sqrt{1-x^2}}{\tau + x} \quad (13)$$

This still leaves the problem of IKIN. However, we can sort that out too. We can find the value of $\theta_{p_{CM}}$ for which $\theta_{p_{Lab}}$ has its maximum by differentiating Eq. 1 and setting it to zero. Once we have this value, we can substitute in Eq. 2 to get the corresponding value of $\theta_{t_{Lab}}$, which we can compare with the value for which we are calculating and from that we can figure out IKIN.

So the problem is to evaluate:

$$\frac{d}{d\theta_{PCM}} \left(\arctan \left(\frac{\sin \theta_{PCM}}{\tau + \cos \theta_{PCM}} \right) \right) = 0 \quad (14)$$

In fact, it turns out that we only need to evaluate:

$$\frac{d}{d\theta_{PCM}} \left(\frac{\sin \theta_{PCM}}{\tau + \cos \theta_{PCM}} \right) = 0 \quad (15)$$

since the arc tangent only introduces a $1/(1+x^2)$ term, which cannot be zero for real values of θ_{PCM} .

let $u = \sin \theta_{PCM}$ and $v = \tau + \cos \theta_{PCM}$.

$$\frac{(\tau + \cos \theta_{PCM}) \left(\frac{d}{d\theta_{PCM}} (\sin \theta_{PCM}) \right) - (\sin \theta_{PCM}) \left(\frac{d}{d\theta_{PCM}} (\tau + \cos \theta_{PCM}) \right)}{(\tau + \cos \theta_{PCM})^2} = 0 \quad (16)$$

or

$$(\tau + \cos \theta_{PCM})(\cos \theta_{PCM}) - (\sin \theta_{PCM})(-\sin \theta_{PCM}) = 0 \quad (17)$$

$$\tau \cos \theta_{PCM} + \cos^2 \theta_{PCM} + \sin^2 \theta_{PCM} = 0 \quad (18)$$

$$\cos \theta_{PCM} = -\frac{1}{\tau} \quad (19)$$

Then we just substitute this value of θ_{PCM} into Eq. 2 and if the value of $\theta_{t_{Lab}}$ that we have to evaluate is greater than this value, we set IKIN = 1, otherwise we set IKIN = 0.

5 What about the other solution?

So far, we have neglected the second solution corresponding to the minus sign in Eq. 12. It is clear that in the specific case of ^{88}Kr that we have considered here, this is not a problem. The lower solution, within the angle ranges of interest correspond to tiny target recoil energies, that probably don't even get out of the target, much less get detected in the particle detector. However, it is not clear that this is generally true.

However, we cannot just add in these events with the second solution, because although they are perfectly valid events, even if in some special case, some of them are detectable, in most cases most of them, if not all of them, will **not** be detectable. However, the Coulomb excitation cross section for these nuclei is very large. In other words, we have a very high probability of Coulomb exciting the nuclei so that the target nuclei have virtually no energy and the projectile is virtually undeviated, but we cannot detect these events, because of their low

energy.

The correct thing to do would be to include these events, but with a gate on the energy, corresponding to the real detection threshold of the system. This could be done, but it would need some extra input, and it is not clear whether such cases will ever really come up.

Should we integrate over both solutions with an energy cut-off or should we just ignore the lower solution?

6 Implementation

As we need to change the input format to give the angle of the detected particle rather than always using that of the scattered projectile, we cannot do this in OP,INTG without breaking compatibility. For this reason, I have chosen to add an OP,INTI which is identical (cut and paste) to the OP,INTG code, except that duplicate FORMAT statements have been removed and duplicate labels have been renumbered and that the new INVKIN subroutine is invoked if the sign of θ_{LAB} is negative.

The new code is in OP,INTI and there are no changes to OP,INTG.

To use it, just replace OP,INTG with OP,INTI and give the laboratory angles of the detected particle for the meshpoints, rather than for the scattered projectile.

7 The INVKIN subroutine

```
C-----  
C SUBROUTINE INVKIN  
C  
C Called by: GOSIA  
C  
C Purpose: calculate the angle of the scattered projectile in the lab frame  
C           when the user gave the angle of the recoiling target nucleus in  
C           the lab frame. There are two solutions to this problem, so Iflag  
C           = 1 selects the larger angle one and Iflag = 2 the smaller one.  
C           Note that the smaller angle (Iflag = 2) corresponds to very low  
C           energies of the recoiling target nucleus, which probably either  
C           don't get out of the target or don't get detected. So Iflag = 2
```

```

C          is probably not very useful! Also, this routine calculates the
C          correct value of the kinematic flag IKIN.
C
C Formal parameters:
C   E_p      - Beam energy in MeV (readonly)
C   E_x      - energy of excited state to use for kinematic in MeV (readonly)
C   M_p      - mass of projectile nuclei in AMU (readonly)
C   M_t      - mass of target nuclei in AMU (readonly)
C   Theta_t  - theta of recoiling target nucleus in lab frame (readonly)
C   Theta_p  - theta of scattered projectile in lab frame (writeonly)
C   Iflag    - flag to select one of two possible solutions (readonly)
C   Ikin     - kinematic flag (writeonly)

SUBROUTINE INVKIN(E_p, E_x , M_p, M_t , Theta_t , Theta_p ,
&                Iflag , Ikin)

REAL*8 E_p , M_p , M_t , Theta_t , Theta_p , E_x
REAL*8 ared , epmin , t , x(2), y , thres
INTEGER*4 Iflag , Ikin

C   Reduced mass

      ared = 1 + M_p / M_t

C   Excitation energy of inelastically scattered particle when state at
C   energy E_x is excited

      epmin = E_p - E_x * ared

C   Tau

      taup = sqrt(E_p / epmin)
      tau = taup * M_p / M_t

C   Calculate the two solutions

      y = tan(theta_t/57.2957795)
      t = taup * taup * y * y * y * y -
&      (1 + y * y) * (taup * taup * y * y - 1)
      t = sqrt(t)
      x(1) = (taup * y * y + t) / (1 + y * y)
      x(1) = atan2(sqrt(1 - x(1) * x(1)), tau + x(1))
      x(2) = (taup * y * y - t) / (1 + y * y)
      x(2) = atan2(sqrt(1 - x(2) * x(2)), tau + x(2))

C   Select the solution we want according to the flag. Note that the

```

```

C      solution with the lower angle corresponds to target recoils which
C      are probably undetectable.

      IF ( Iflag.EQ.1 ) THEN
          theta_p = MAX(x(1),x(2))*57.2957795
      ELSE
          theta_p = MIN(x(1),x(2))*57.2957795
      ENDIF

C      Calculate angle of scattered projectile in centre of mass frame, for
C      which the maximum laboratory scattering angle is reached.

      t = acos(-1./tau)

C      Now calculate the arctangent of the corresponding angle for the
C      recoiling target nuclei in the laboratory frame

      thres = sin(t)/(taup-cos(t))

C      So now, if  $y = \tan(\theta_{t\_lab}) > thres$ , we are above the maximum and
C      need the larger value of  $\theta_{p\_cm}$ , so we set  $I_{kin}$  to 1. Otherwise we
C      are below the maximum and need the smaller value so we choose  $I_{kin} = 0$ .

      IF ( y.GT.thres ) THEN
          Ikin = 1
      ELSE
          Ikin = 0
      ENDIF

      END

```

8 Tests

Several tests have been made with this code.

8.1 OP,INTG check

As a first test, OP,INTG was run with the old and new code using an energy range from 90.3 to 193.6 MeV, with energy meshpoints 90, 105, 120, 135, 150, 165, 180 and 195 MeV and θ meshpoints -4.36° , -5.50° , -6.42° , -7.10° and -7.51° integrating from 16° to 34° in the target recoil angle.

These meshpoints correspond to, which correspond to 15° , 20° , 25° , 30° and 35° for the target recoil at 142 MeV, which is the average energy in the target.

This gave exactly identical results. This is not surprising, since OP,INTG is unchanged in the new code.

8.2 OP,INTI check

The same test was performed, but with the new code using the OP,INTI option and the angles given as target recoil angles.

Here, we see that the new code is giving a different set of angles for each energy, but the same “RESPECTIVE TARGET SCATTERING ANGLE”:

CALCULATED YIELDS

```

EXPERIMENT 1 DETECTOR 1
ENERGY      90.000 MEV THETA  -4.310 DEG

NI      NF      II      IF      YIELD      NORMALIZED YIELD

RESPECTIVE TARGET SCATTERING ANGLE= 15.000 DEG

 3       2       4.0       2.0       0.56298E-07       0.29934E-06
 2       1       2.0       0.0       0.18807E+00       0.10000E+01

```

and

CALCULATED YIELDS

```

EXPERIMENT 1 DETECTOR 1
ENERGY     120.000 MEV THETA  -4.338 DEG

NI      NF      II      IF      YIELD      NORMALIZED YIELD

RESPECTIVE TARGET SCATTERING ANGLE= 15.000 DEG

 3       2       4.0       2.0       0.34636E-04       0.15518E-04
 2       1       2.0       0.0       0.22319E+01       0.10000E+01

```

Note how the “RESPECTIVE TARGET SCATTERING ANGLE” is the same in both, but the THETA angle is different. It is the former that is given in the input for OP,INTI in the new code and the THETA angle is calculated. What is printed in the output is actually, what gosia calculates, when it goes back from the calculated THETA to the target angle using the original code. This should and does agree with the value in the input.

Quantity	Old Code	New Code	Difference
Rutherford cross section	2564	2596	0.1 %
Yield 1 st excited state	2.1221	2.1261	0.2 %
Yield 2 nd excited state	7.0565×10^{-4}	7.0338×10^{-4}	0.3 %

Although there is a small numerical difference, this is not significant. A difference is inevitable, since we are using different meshpoints.

8.3 Test with other IKIN

The above tests were with $IKIN = 0$. The next test is with $IKIN = 1$ over angles from 51° to 69° for the same energy range and energy meshpoints.

For the new code, the meshpoints were given as 47° , 51° , 54° , 57° , 60° , 63° , 66° , 69° and 71° .

For the old code, the corresponding values for the scattered projectile angle were given evaluated at the mean energy: 7.46° , 7.24° , 6.95° , 6.58° , 6.15° , 5.65° , 5.10° , 4.48° and 3.79° .

This gives:

Quantity	Old Code	New Code	Difference
Rutherford cross section	53900	47390	12.1 %
Yield 1 st excited state	4.1988	4.4144	5.1 %
Yield 2 nd excited state	2.8967×10^{-4}	6.7095×10^{-4}	132 %

So there are huge discrepancies here. However, this turns out to be because we are integrating over the large target-recoil angles, which correspond to the very small angles of the scattered projectile, where the Rutherford cross section is very large and, more importantly, changes rapidly. Since the two codes are effectively using different angles for the meshpoints, it is not surprising that they differ, however this difference is not normal. Further investigation shows that it is due to the way the lagrangian interpolation works (or fails to work). Turning on Pawel's spline with the same input gives:

Quantity	Old Code	New Code	Difference
Rutherford cross section	47520	47460	0.1 %
Yield 1 st excited state	4.3508	4.3876	0.8 %
Yield 2 nd excited state	3.8543×10^{-4}	3.8885×10^{-4}	0.9%

The spline option is important for large target recoil angles.

8.4 Using both IKIN with a single energy

In order to test the way the new code selects between $IKIN = 0$ and $IKIN = 1$, we can test with a single energy meshpoint, so the energy effect is neglected. This is not a realistic calculation of the Coulomb excitation, but it allows us to make a direct comparison.

For the old code, we need to split the θ range into a low and high part. The low part runs from 16.4° to the point where we need to switch from $\text{IKIN} = 0$ to $\text{IKIN} = 1$ and the high part runs from that point to 53.0° . For the 0.775 MeV state this is at 40.72279° .

So for the low part we use meshpoints at 4.37° , 5.52° , 6.45° , 7.13° , 7.54° and 7.70° with limits 16.4° to 40.72° .

Quantity	Old Code (low)	Old Code (high)	Old Code (sum)	New Code	Difference
Rutherford	2053	3296	5349	5346	0.1 %
Yld 1 st ex. state	8.8048	6.9624	15.7672	15.746	0.1 %
Yld 2 nd ex. state $\times 10^3$	5.8464	2.9647	8.8111	8.7996	0.1 %

It seems to work, but we need some more intensive testing...

9 And the next problem...?

At the Miniball workshop it was decided to use the new target chamber developed for transfer experiments also for Coulomb excitation experiments. This target chamber has a barrel-shaped particle detector, so the angular coverage is from about 16.4° to backward angles.

Suppose a ^{88}Kr nucleus, with an initial energy of 193.6 MeV, is incident on a 2.1 mg/cm^2 ^{12}C target and a Coulomb excitation interaction occurs at the middle of the target, so that the ^{12}C nucleus is scattered to 60° in the laboratory frame. The incoming ^{88}Kr ion will lose about 51.5 MeV in the target and have about 142.1 MeV when it interacts. For this energy, when the ^{12}C target nucleus scatters to 60° , it will have an energy of about 13.60 MeV. At this angle, it will have to traverse a thickness of 2.1 mg/cm^2 of ^{12}C . The energy loss is then about 12.95 MeV, so the ion which hits the particle detector only has an energy of about 0.65 MeV and will probably not be detected.

So these events will not contribute to the measured cross-section, but Gosia will still include them in the integration.

Unfortunately, figuring out which points to include and which not turns out to be non-trivial. I thought at first, that we could just set an energy threshold, but it turns out that the loss of energy in the target is very important for these low energies. So we would need to take the stopping power into account, and this is not the same stopping power that Gosia already has, but the stopping power for target recoils in the target.

Any thoughts about this?